
AshGraphql
Welcome! This is the extension for building GraphQL APIs with Ash. The generated GraphQL APIs are powered by Absinthe. Generate a powerful Graphql API in minutes!

Getting Started With GraphQL
    


  
    
  
  Get familiar with Ash resources


If you haven't already, read the Ash Getting Started Guide. This assumes that you already have resources set up, and only gives you the steps to add AshGraphql to your resources/domains.

  
    
  
  Installation



  
    
  
  Using Igniter (recommended)


mix igniter.install ash_graphql

  
    
  
  Manual


Bring in the ash_graphql dependency
def deps()
  [
    ...
    {:ash_graphql, "~> 1.4.7"}
  ]
end
Setting up your schema
If you don't have an absinthe schema, you can create one just for ash.
Replace helpdesk in the examples with your own application name.
See the SDL file guide for more information on using the SDL file,
or remove the generate_sdl_file option to skip generating it on calls to mix ash.codegen.
in lib/helpdesk/schema.ex
defmodule Helpdesk.GraphqlSchema do
  use Absinthe.Schema

  # Add your domains here
  use AshGraphql,
    domains: [Your.Domains]

  query do
    # Custom absinthe queries can be placed here
    @desc "Remove me once you have a query of your own!"
    field :remove_me, :string do
      resolve fn _, _, _ ->
        {:ok, "Remove me!"}
      end
    end
  end

  mutation do
    # Custom absinthe mutations can be placed here
  end
end
Connect your schema
Using Phoenix
Add the following code to your Phoenix router. It's useful to set up the Absinthe playground for trying things out, but it's optional.
pipeline :graphql do
  plug AshGraphql.Plug
end

scope "/gql" do
  pipe_through [:graphql]

  forward "/playground",
          Absinthe.Plug.GraphiQL,
          schema: Module.concat(["Helpdesk.GraphqlSchema"]),
          interface: :playground

  forward "/",
    Absinthe.Plug,
    schema: Module.concat(["Helpdesk.GraphqlSchema"])
end

  
    
  
  Whats up with Module.concat/1?


This Module.concat/1 prevents a compile-time dependency from this router module to the schema module. It is an implementation detail of how forward/2 works that you end up with a compile-time dependency on the schema, but there is no need for this dependency, and that dependency can have drastic impacts on your compile times in certain scenarios.

If you started with mix new ... instead of mix phx.new ... and you want to
still use Phoenix, the fastest path that way is typically to just create a new
Phoenix application and copy your resources/config over.
Using Plug
If you are unfamiliar with how plug works, this guide
will be helpful for understanding it. It also guides you through adding plug to your application.
Then you can use a Plug.Router and forward to your plugs similar to how it is done for phoenix:
plug AshGraphql.Plug

forward "/gql",
  to: Absinthe.Plug,
  init_opts: [schema: Module.concat(["Helpdesk.GraphqlSchema"])]

forward "/playground",
  to: Absinthe.Plug.GraphiQL,
  init_opts: [
    schema: Module.concat(["Helpdesk.GraphqlSchema"]),
    interface: :playground
  ]
For information on why we are using Module.concat/1, see the note above in the Phoenix section.

  
    
  
  Select domains to show in your GraphQL


In the use AshGraphql call in your schema, you specify which domains you want to expose in your GraphQL API. Add any domains that will have AshGraphql queries/mutations to the domains list. For example:
use AshGraphql, domains: [Your.Domain1, Your.Domain2]

  
    
  
  Adding Queries and Mutations


Some example queries/mutations are shown below. If no queries/mutations are added, nothing will show up in the GraphQL API, so be sure to set one up if you want to try it out.

  
    
  
  Queries & Mutations on the Resource


Here we show queries and mutations being added to the resource, but you can also define them on the domain. See below for an equivalent definition
defmodule Helpdesk.Support.Ticket do
  use Ash.Resource,
    ...,
    extensions: [
      AshGraphql.Resource
    ]

  graphql do
    type :ticket

    queries do
      # Examples

      # create a field called `get_ticket` that uses the `read` read action to fetch a single ticke
      get :get_ticket, :read
      # create a field called `most_important_ticket` that uses the `most_important` read action to fetch a single record
      read_one :most_important_ticket, :most_important

      # create a field called `list_tickets` that uses the `read` read action to fetch a list of tickets
      list :list_tickets, :read
    end

    mutations do
      # Examples

      create :create_ticket, :create
      update :update_ticket, :update
      destroy :destroy_ticket, :destroy
    end
  end

  ...
end

  
    
  
  Queries & Mutations on the Domain


defmodule Helpdesk.Support.Ticket do
  use Ash.Resource,
    ...,
    extensions: [
      AshGraphql.Resource
    ]

  # The resource still determines its type, and any other resource/type-based
  # configuration
  graphql do
    type :ticket
  end

  ...
end

defmodule Helpdesk.Support do
  use Ash.Domain,
    extensions: [
      AshGraphql.Domain
    ]

  ...
  graphql do
    # equivalent queries and mutations, but the first argument
    # is the resource because the domain can define queries for
    # any of its resources
    queries do
      get Helpdesk.Support.Ticket, :get_ticket, :read
      read_one Helpdesk.Support.Ticket, :most_important_ticket, :most_important
      list Helpdesk.Support.Ticket, :list_tickets, :read
    end

    mutations do
      create Helpdesk.Support.Ticket, :create_ticket, :create
      update Helpdesk.Support.Ticket, :update_ticket, :update
      destroy Helpdesk.Support.Ticket, :destroy_ticket, :destroy
    end
end

  
    
  
  What's next?


Topics:
	GraphQL Generation

How Tos:
	Authorize With GraphQL
	Handle Errors
	Use Enums with GraphQL
	Use JSON with GraphQL




  

    
Authorize with GraphQL
    

AshGraphql uses three special keys in the absinthe context:
	:actor - the current actor, to be used for authorization/preparations/changes
	:tenant - a tenant when using multitenancy.
	:ash_context - a map of arbitrary context to be passed into the changeset/query. Accessible via changeset.context and query.context

By default, authorize? in the domain is set to true. To disable authorization for a given domain in graphql, use:
graphql do
  authorize? false
end
If you are doing authorization, you'll need to provide an actor.
To set the actor for authorization, you'll need to add an actor key to the
absinthe context. Typically, you would have a plug that fetches the current user and uses Ash.PlugHelpers.set_actor/2 to set the actor in the conn (likewise with Ash.PlugHelpers.set_tenant/2).
Just add AshGraphql.Plug somewhere after that in the pipeline and the your
GraphQL APIs will have the correct authorization.
defmodule MyAppWeb.Router do
  pipeline :api do
    # ...
    plug :get_actor_from_token
    plug AshGraphql.Plug
  end

  scope "/" do
    forward "/gql", Absinthe.Plug, schema: YourSchema

    forward "/playground",
          Absinthe.Plug.GraphiQL,
          schema: YourSchema,
          interface: :playground
  end

  def get_actor_from_token(conn, _opts) do
     with ["" <> token] <- get_req_header(conn, "authorization"),
         {:ok, user, _claims} <- MyApp.Guardian.resource_from_token(token) do
      conn
      |> set_actor(user)
    else
    _ -> conn
    end
  end
end

  
    
  
  Policy Breakdowns


By default, unauthorized requests simply return forbidden in the message. If you prefer to show policy breakdowns in your GraphQL errors, you can set the config option:
config :ash_graphql, :policies, show_policy_breakdowns?: true
{
  "data": {
    "attendanceRecords": null
  },
  "errors": [
    {
      "code": "forbidden",
      "fields": [],
      "locations": [
        {
          "column": 3,
          "line": 2
        }
      ],
      "message": "MyApp.Authentication.User.read\n\n\n\n\nPolicy Breakdown\n  Policy | ⛔:\n    forbid unless: actor is active | ✓ | ⬇    \n    authorize if: actor is Executive | ✘ | ⬇",
      "path": ["attendanceRecords"],
      "short_message": "forbidden",
      "vars": {}
    }
  ]
}
Be careful, as this can be an attack vector in some systems (i.e "here is exactly what you need to make true to do what you want to do").

  
    
  
  Field Policies


Field policies in AshGraphql work by producing a null value for any forbidden field, as well as an error in the errors list.

  
    
  
  nullability


Any fields with field policies on them should be nullable. If they are not nullable, the parent object will also be null (and considered in an error state), because null is not a valid type for that field.

To make fields as nullable even if it is not nullable by its definition, use the nullable_fields option.
graphql do
  type :post

  nullable_fields [:foo, :bar, :baz]
end



  

    
Handling Errors
    

There are various options that can be set on the Domain module to determine how errors behave and/or are shown in the GraphQL.

  
    
  
  Showing raised errors


For security purposes, if an error is raised as opposed to returned somewhere, the error is hidden. Set this to true in dev/test environments for an easier time debugging.
graphql do
  show_raised_errors? true
end

# or it can be done in config
# make sure you've set `otp_app` in your domain, i.e use Ash.Domain, otp_app: :my_app

config :my_app, YourDomain, [
  graphql: [
    show_raised_errors?: true
  ]
]

  
    
  
  Root level errors


By default, action errors are simply shown in the errors field for mutations. Set this to true to return them as root level errors instead.
graphql do
  root_level_errors? true
end

  
    
  
  Error Handler


Setting an error handler allows you to use things like gettext to translate errors and/or modify errors in some way. This error handler will take the error object to be returned, and the context. See the absinthe docs for adding to the absinthe context (i.e for setting a locale).
graphql do
  error_handler {MyApp.GraphqlErrorHandler, :handle_error, []}
end
Keep in mind, that you will want to ensure that any custom error handler you add performs the logic to replace variables in error messages. 
This is what the default error handler looks like, for example:
defmodule AshGraphql.DefaultErrorHandler do
  @moduledoc "Replaces any text in message or short_message with variables"

  def handle_error(
        %{message: message, short_message: short_message, vars: vars} = error,
        _context
      ) do
    %{
      error
      | message: replace_vars(message, vars),
        short_message: replace_vars(short_message, vars)
    }
  end

  def handle_error(other, _), do: other

  defp replace_vars(string, vars) do
    vars =
      if is_map(vars) do
        vars
      else
        List.wrap(vars)
      end

    Enum.reduce(vars, string, fn {key, value}, acc ->
      if String.contains?(acc, "%{#{key}}") do
        String.replace(acc, "%{#{key}}", to_string(value))
      else
        acc
      end
    end)
  end
end

  
    
  
  Custom Errors


If you created your own Errors as described in the Ash Docs you also need to implement
the protocol for it to be displayed in the Api.
defmodule Ash.Error.Action.InvalidArgument do
  @moduledoc "Used when an invalid value is provided for an action argument"
  use Splode.Error, fields: [:field, :message, :value], class: :invalid

  def message(error) do
    """
    Invalid value provided#{for_field(error)}#{do_message(error)}

    #{inspect(error.value)}
    """
  end
  
  defimpl AshGraphql.Error, for: Ash.Error.Changes.InvalidArgument do
    def to_error(error) do
      %{
        message: error.message,
        short_message: error.message,
        code: "invalid_argument",
        vars: Map.new(error.vars),
        fields: [error.field]
      }
    end
  end
end



  

    
Generic Actions
    

Generic actions allow us to build any interface we want in Ash. AshGraphql
has full support for generic actions, from type generation to data loading.
This means that you can write actions that return records or lists of records
and those will have all of their fields appropriately loadable, or you can have
generic actions that return simple scalars, like integers or strings.

  
    
  
  Examples


Here we have a simple generic action returning a scalar value.
graphql do
  queries do
    action :say_hello, :say_hello
  end
end

actions do
  action :say_hello, :string do
    argument :to, :string, allow_nil?: false

    run fn input, _ ->
      {:ok, "Hello, #{input.arguments.to}"}
    end
  end
end
And here we have a generic action returning a list of records.
graphql do
  type :post

  queries do
    action :random_ten, :random_ten
  end
end

actions do
  action :random_ten, {:array, :struct} do
    constraints items: [instance_of: __MODULE__]

    run fn input, context ->
      # This is just an example, not an efficient way to get
      # ten random records
      with {:ok, records} <-  Ash.read(__MODULE__) do
        {:ok, Enum.take_random(records, 10)}
      end
    end
  end
end



  

    
Using the SDL File
    

By passing the generate_sdl_file to use AshGraphql, AshGraphql will generate
a schema file when you run mix ash.codegen. For example:
use AshGraphql,
  domains: [Domain1, Domain2],
  generate_sdl_file: "priv/schema.graphql"

  
    
  
  Ensure your schema is up to date, gitignored, or not generated


We suggest first adding mix ash.codegen --check to your CI/CD pipeline to
ensure the schema is always up-to-date. Alternatively you can add the file
to your .gitignore, or you can remove the generate_sdl_file option to skip
generating the file.

With the generate_sdl_file option, calls to mix ash.codegen <name> will generate
a .graphql file at the specified path.
Some things that you can use this SDL file for:

  
    
  
  Documentation


The schema file itself represents your entire GraphQL API definition, and examining it can be very useful.

  
    
  
  Code Generation


You can use tools like GraphQL codegen to generate a client
for your GraphQL API.

  
    
  
  Validating Changes


Use the SDL file to check for breaking changes in your schema, especially if you are exposing a public API.
A plug and play github action for this can be found here: https://the-guild.dev/graphql/inspector/docs/products/action



  

    
Use Enums with GraphQL
    

If you define an Ash.Type.Enum, that enum type can be used both in attributes and arguments. You will need to add graphql_type/0 to your implementation. AshGraphql will ensure that a single type is defined for it, which will be reused across all occurrences. If an enum
type is referenced, but does not have graphql_type/0 defined, it will
be treated as a string input.
For example:
defmodule AshPostgres.Test.Types.Status do
  @moduledoc false
  use Ash.Type.Enum, values: [:open, :closed]

  def graphql_type(_), do: :ticket_status

  # Optionally, remap the names used in GraphQL, for instance if you have a value like `:"10"`
  # that value is not compatible with GraphQL

  def graphql_rename_value(:"10"), do: :ten
  def graphql_rename_value(value), do: value

  # You can also provide descriptions for the enum values, which will be exposed in the GraphQL
  # schema.
  # Remember to have a fallback clause that returns nil if you don't provide descriptions for all
  # values.

  def graphql_describe_enum_value(:open), do: "The post is open"
  def graphql_describe_enum_value(_), do: nil
end


  
    
  
  Using custom absinthe types


You can implement a custom enum by first adding the enum type to your absinthe schema (more here). Then you can define a custom Ash type that refers to that absinthe enum type.
# In your absinthe schema:

enum :status do
  value(:open, description: "The post is open")
  value(:closed, description: "The post is closed")
end
# Your custom Ash Type
defmodule AshGraphql.Test.Status do
  use Ash.Type.Enum, values: [:open, :closed]

  use AshGraphql.Type

  @impl true
  # tell Ash not to define the type for that enum
  def graphql_define_type?(_), do: false
end



  

    
Use JSON with GraphQL
    

AshGraphql provides two JSON types that may be used. They are the same except for how the type is serialized in responses.
	:json_string - serializes the json to a string, e.g "{\"foo\":1}"
	:json - leaves the json as an object, e.g {foo: 1}

By default, :json_string is used. The configuration for this is (uncharacteristically) placed in application config, for example:
config :ash_graphql, :json_type, :json



  

    
Using Subscriptions
    

You can do this with Absinthe directly, and use
AshGraphql.Subscription.query_for_subscription/3. Here is an example of how
you could do this for a subscription for a single record. This example could be
extended to support lists of records as well.
# in your absinthe schema file
subscription do
  field :field, :type_name do
    config(fn
      _args, %{context: %{current_user: %{id: user_id}}} ->
        {:ok, topic: user_id, context_id: "user/#{user_id}"}

      _args, _context ->
        {:error, :unauthorized}
    end)

    resolve(fn args, _, resolution ->
      # loads all the data you need
      AshGraphql.Subscription.query_for_subscription(
        YourResource,
        YourDomain,
        resolution
      )
      |> Ash.Query.filter(id == ^args.id)
      |> Ash.read(actor: resolution.context.current_user)
    end)
  end
end

  
    
  
  Subscription DSL (beta)


The subscription DSL is currently in beta and before using it you have to enable
them in your config.

  
    
  
  Subscription response order


The order in which the subscription responses are sent to the client is not
guaranteed to be the same as the order in which the mutations were executed.

config :ash_graphql, :subscriptions, true
First you'll need to do some setup, follow the the
setup guide
in the absinthe docs, but instead of using Absinthe.Pheonix.Endpoint use
AshGraphql.Subscription.Endpoint.
By default subscriptions are resolved synchronously as part of the mutation.
This means that a resolver is run for every subscriber that is not deduplicated.
If you have a lot of subscribers you can add the
AshGraphql.Subscription.Batcher to your supervision tree, which batches up
notifications and runs subscription resolution out-of-band.
  @impl true
  def start(_type, _args) do
    children = [
      ...,
      {Absinthe.Subscription, MyAppWeb.Endpoint},
      AshGraphql.Subscription.Batcher
    ]

    # See https://hexdocs.pm/elixir/Supervisor.html
    # for other strategies and supported options
    opts = [strategy: :one_for_one, name: MyAppWeb.Supervisor]
    Supervisor.start_link(children, opts)
  end
Afterwards, add an empty subscription block to your schema module.
defmodule MyAppWeb.Schema do
  ...

  subscription do
  end
end
Now you can define subscriptions on your resource or domain
defmodule MyApp.Resource do
  use Ash.Resource,
  data_layer: Ash.DataLayer.Ets,
  extensions: [AshGraphql.Resource]

  graphql do
    subscriptions do
      subscribe :resource_created do
        action_types :create
      end
    end
  end
end
For further Details checkout the DSL docs for
resource
and domain

  
    
  
  Deduplication


By default, Absinthe will deduplicate subscriptions based on the context_id.
We use the some of the context like actor and tenant to create a context_id
for you.
If you want to customize the deduplication you can do so by adding a actor
function to your subscription. This function will be called with the actor that
subscribes and you can return a more generic actor, this way you can have one
actor for multiple users, which will lead to less resolver executions.
defmodule MyApp.Resource do
  use Ash.Resource,
  data_layer: Ash.DataLayer.Ets,
  extensions: [AshGraphql.Resource]

  graphql do
    subscriptions do
      subscribe :resource_created do
        action_types :create
        actor fn actor ->
          if check_actor(actor) do
            %{id: "your generic actor", ...}
          else
            actor
          end
        end
      end
    end
  end
end



  

    
Use Unions with GraphQL
    

Unions must be defined with Ash.Type.NewType:
defmodule MyApp.Armor do
  use Ash.Type.NewType, subtype_of: :union, constraints: [
    types: [
      plate: [
        # This is an embedded resource, with its own fields
        type: MyApp.Armor.Plate
      ],
      chain_mail: [
        # And so is this
        type: MyApp.Armor.ChainMail
      ],
      custom: [
        type: :string
      ]
    ]
  ]

  use AshGraphql.Type

  # Add this to define the union in ash_graphql
  def graphql_type(_), do: :armor
end
By default, the type you would get for this on input and output would look something like this:
type Armor = {plate: {value: Plate}} | {chain_mail: {value: ChainMail}} | {custom: {value: String}}
We do this by default to solve for potentially ambiguous types. An example of this might be if you had multiple different types of strings in a union, and you wanted the client to be able to tell exactly which type of string they'd been given. i.e {social: {value: "555-55-5555"}} | {phone_number: {value: "555-5555"}}.
However, you can clean the type in cases where you have no such conflicts by by providing
# Put anything in here that does not need to be named/nested with `{type_name: {value: value}}`
def graphql_unnested_unions(_constraints), do: [:plate, :chain_mail]
Which, in this case, would yield:
type Armor = Plate | ChainMail | {custom: {value: String}}

  
    
  
  Bypassing type generation for a union


Add the graphql_define_type?/1 callback, like so, to skip Ash's generation (i.e if you're defining it yourself)
@impl true
def graphql_define_type?(_), do: false



  

    
Use Maps with GraphQL
    

If you define an Ash.Type.NewType that is a subtype of :map, and you add the fields constraint which specifies field names and their types, AshGraphql will automatically derive an appropriate GraphQL type for it.
For example:
defmodule MyApp.Types.Metadata do
  @moduledoc false
  use Ash.Type.NewType, subtype_of: :map, constraints: [
    fields: [
      title: [
        type: :string
      ],
      description: [
        type: :string
      ]
    ]
  ]

  def graphql_type(_), do: :metadata
end


  
    
  
  Bypassing type generation for an map


Add the graphql_define_type?/1 callback, like so, to skip Ash's generation (i.e if you're defining it yourself)
@impl true
def graphql_define_type?(_), do: false



  

    
Monitoring
    

Please read the Ash monitoring guide for more information. Here we simply cover the additional traces & telemetry events that we publish from this extension.
A tracer can be configured in the domain. It will fallback to the global tracer configuration config :ash, :tracer, Tracer
graphql do
  trace MyApp.Tracer
end

  
    
  
  Traces


Each graphql resolver, and batch resolution of the underlying data loader, will produce a span with an appropriate name. We also set a source: :graphql metadata if you want to filter them out or annotate them in some way.

  
    
  
  Telemetry


AshGraphql emits the following telemetry events, suffixed with :start and :stop. Start events have system_time measurements, and stop events have system_time and duration measurements. All times will be in the native time unit.
	[:ash, <domain_short_name>, :gql_mutation] - The execution of a mutation. Use resource_short_name and mutation (or action) metadata to break down measurements.

	[:ash, <domain_short_name>, :gql_query] - The execution of a mutation. Use resource_short_name and query (or action) metadata to break down measurements.

	[:ash, <domain_short_name>, :gql_relationship] - The resolution of a relationship. Use resource_short_name and relationship metadata to break down measurements.

	[:ash, <domain_short_name>, :gql_calculation] - The resolution of a calculation. Use resource_short_name and calculation metadata to break down measurements.

	[:ash, <domain_short_name>, :gql_relationship_batch] - The resolution of a batch of relationships by the data loader. Use resource_short_name and relationship metadata to break down measurements.

	[:ash, <domain_short_name>, :gql_calculation_batch] - The resolution of a batch of calculations by the data loader. Use resource_short_name and calculation metadata to break down measurements.





  

    
GraphQL Query Generation
    

Following where we left off from Getting Started with GraphQL, this guide explores what the GraphQL requests and responses look like for different queries defined with the AshGraphql DSL.
All of the following examples apply to queries & mutations places on the domain as well.

  
    
  
  Fetch Data by ID


defmodule Helpdesk.Support.Ticket do
  use Ash.Resource,
    ...,
    extensions: [
      AshGraphql.Resource
    ]

  attributes do
    # Add an autogenerated UUID primary key called `:id`.
    uuid_primary_key :id

    # Add a string type attribute called `:subject`
    attribute :subject, :string
  end

  actions do
    # Add a set of simple actions. You'll customize these later.
    defaults [:read, :update, :destroy]
  end

  graphql do
    type :ticket

    queries do
      # create a field called `get_ticket` that uses the `read` read action to fetch a single ticket
      get :get_ticket, :read
    end
  end
end
For the get_ticket query defined above, the corresponding GraphQL would look like this:
query ($id: ID!) {
  getTicket(id: $id) {
    id
    subject
  }
}
And the response would look similar to this:
{
  "data": {
    "getTicket": {
      "id": "",
      "subject": ""
    }
  }
}
Let's look at an example of querying a list of things.
  graphql do
    type :ticket

    queries do
      # create a field called `get_ticket` that uses the `read` read action to fetch a single ticket
      get :get_ticket, :read

      # create a field called `list_tickets` that uses the `read` read action to fetch a list of tickets
      list :list_tickets, :read
    end
  end
This time, we've added list :list_tickets, :read, to generate a GraphQL query for listing tickets.
The request would look something like this:
query {
  listTickets {
    id
    subject
  }
}
And the response would look similar to this:
{
  "data": {
    "listTickets": [
      {
        "id": "",
        "subject": ""
      }
    ]
  }
}

  
    
  
  Filter Data With Arguments


Now, let's say we want to add query parameters to listTickets. How do we do that?
Consider list :list_tickets, :read and the actions section:
  actions do
    # Add a set of simple actions. You'll customize these later.
    defaults [:read, :update, :destroy]
  end

  graphql do
    type :ticket

    queries do
      # create a field called `list_tickets` that uses the `read` read action to fetch a list of tickets
      list :list_tickets, :read
    end
  end
The second argument to list :list_tickets, :read is the action that will be called when the query is run.
In the current example, the action is :read, which is the generic Read action.
Let's create a custom action in order to define query parameters for the listTickets query.
We'll call this action :query_tickets:
  actions do
    defaults [:read, :update, :destroy]

    read :query_tickets do
      argument :representative_id, :uuid

      filter(
        expr do
          is_nil(^arg(:representative_id)) or representative_id == ^arg(:representative_id)
        end
      )
    end
  end

  graphql do
    type :ticket

    queries do
      # create a field called `list_tickets` that uses the `:query_tickets` read action to fetch a list of tickets
      list :list_tickets, :query_tickets
    end
  end
In the graphql section, the list/2 call has been changed, replacing the :read action with :query_tickets.
The GraphQL request would look something like this:
query ($representative_id: ID) {
  list_tickets(representative_id: $representative_id) {
    id
    representative_id
    subject
  }
}

  
    
  
  Mutations and Enums


Now, let's look at how to create a ticket by using a GraphQL mutation.
Let's say you have a Resource that defines an enum-like attribute:
defmodule Helpdesk.Support.Ticket do
  use Ash.Resource,
    ...,
    extensions: [
      AshGraphql.Resource
    ]


  attributes do
    uuid_primary_key :id
    attribute :subject, :string
    attribute :status, :atom, constraints: [one_of: [:open, :closed]]
  end

  actions do
    defaults [:create, :read, :update, :destroy]
  end

  graphql do
    type :ticket

    queries do
      get :get_ticket, :read
    end

    mutations do
      create :create_ticket, :create
    end
  end
end
Above, the following changes have been added:
	In the attributes section, the :status attribute has been added.
	In the actions section, the :create action has been added.
	The :create_ticket mutation has been defined in the new graphql.mutations section.

The :status attribute is an enum that is constrained to the values [:open, :closed].
When used in conjunction with AshGraphql, a GraphQL enum type called TicketStatus will be generated for this attribute.
The possible GraphQL values for TicketStatus are OPEN and CLOSED.
See Use Enums with GraphQL for more information.
We can now create a ticket with the createTicket mutation:
mutation ($input: CreateTicketInput!) {
  createTicket(input: $input) {
    result {
      id
      subject
      status
    }
    errors {
      code
      fields
      message
      shortMessage
      vars
    }
  }
}
Note
	The resulting ticket data is wrapped in AshGraphql's result object.

	Validation errors are wrapped in a list of error objects under errors, also specified in the query.
AshGraphql does this by default instead of exposing errors in GraphQL's standard errors array.
This behavior can be changed by setting root_level_errors? true in the graphql section
of your Ash domain module:
defmodule Helpdesk.Support do
  use Ash.Domain, extensions: [AshGraphql.Domain]

  graphql do
    root_level_errors? true
  end
end


If we were to run this mutation in a test, it would look something like this:
input = %{
  subject: "My Ticket",
  status: "OPEN"
}

resp_body =
  post(conn, "/api/graphql", %{
    query: query,
    variables: %{input: input}
  })
  |> json_response(200)
Notice that the status attribute is set to "OPEN" and not "open". It is important that the value of the status be uppercase.
This is required by GraphQL enums. AshGraphql will automatically convert the value to the correct case.
The response will look something like this:
{
  "data": {
    "createTicket": {
      "result": {
        "id": "b771e433-0979-4d07-a280-4d12373849aa",
        "subject": "My Ticket",
        "status": "OPEN"
      }
    }
  }
}
Again, AshGraphql will automatically convert the status value from :open to "OPEN".

  
    
  
  More GraphQL Docs


If you haven't already, please turn on the documentation tag for AshGraphql. Tags can be controlled
at the top of the left navigation menu, under "Including Libraries:".
	Getting Started With GraphQL
	AshGraphql.Domain




  

    
Modifying the Resolution
    

Using the modify_resolution option, you can alter the Absinthe.Resolution.
modify_resolution is an MFA that will be called with the resolution, the query, and the result of the action as the first three arguments. Must return a new Absinthe.Resolution.
This can be used to implement things like setting cookies based on resource actions. A method of using resolution context for that is documented in Absinthe.Plug

  
    
  
  as_mutation?


If you are modifying the context in a query, then you should also set as_mutation? to true and represent this in your graphql as a mutation. See as_mutation? for more.




  

    
Relay
    

Enabling Relay for a resource sets it up to follow the Relay specification.
The two changes that are made currently are:
	the type for the resource will implement the Node interface
	pagination over that resource will behave as a Connection.


  
    
  
  Using Ash's built-in Relay support


Set relay? true on the resource:
graphql do
  relay? true

  ...
end

  
    
  
  Relay Global IDs


Use the following option to generate Relay Global IDs (see
here).
use AshGraphql, relay_ids?: true
This allows refetching a node using the node query and passing its global ID.

  
    
  
  Translating Relay Global IDs passed as arguments


When relay_ids?: true is passed, users of the API will have access only to the global IDs, so they
will also need to use them when an ID is required as argument. You actions, though, internally use the
normal IDs defined by the data layer.
To handle the translation between the two ID domains, you can use the relay_id_translations
option. With this, you can define a list of arguments that will be translated from Relay global IDs
to internal IDs.
For example, if you have a Post resource with an action to create a post associated with an
author:
create :create do
  argument :author_id, :uuid

  # Do stuff with author_id
end
You can add this to the mutation connected to that action:
mutations do
  create :create_post, :create do
    relay_id_translations [input: [author_id: :user]]
  end
end

  
    
  
  Using with Absinthe.Relay instead of Ash's relay type


Use the following option when calling use AshGraphql
use AshGraphql, define_relay_types?: false

